creation_sourcecode: // Sources flattened with hardhat v2.22.3 https://hardhat.org
  
  // SPDX-License-Identifier: GPL-3.0 AND LGPL-3.0-only AND MIT
  
  // File @account-abstraction/contracts/core/Helpers.sol@v0.6.0
  
  // Original license: SPDX_License_Identifier: GPL-3.0
  pragma solidity ^0.8.12;
  
  /* solhint-disable no-inline-assembly */
  
  /**
   * returned data from validateUserOp.
   * validateUserOp returns a uint256, with is created by `_packedValidationData` and parsed by `_parseValidationData`
   * @param aggregator - address(0) - the account validated the signature by itself.
   *              address(1) - the account failed to validate the signature.
   *              otherwise - this is an address of a signature aggregator that must be used to validate the signature.
   * @param validAfter - this UserOp is valid only after this timestamp.
   * @param validaUntil - this UserOp is valid only up to this timestamp.
   */
  struct ValidationData {
      address aggregator;
      uint48 validAfter;
      uint48 validUntil;
  }
  
  //extract sigFailed, validAfter, validUntil.
  // also convert zero validUntil to type(uint48).max
  function _parseValidationData(
      uint validationData
  ) pure returns (ValidationData memory data) {
      address aggregator = address(uint160(validationData));
      uint48 validUntil = uint48(validationData >> 160);
      if (validUntil == 0) {
          validUntil = type(uint48).max;
      }
      uint48 validAfter = uint48(validationData >> (48 + 160));
      return ValidationData(aggregator, validAfter, validUntil);
  }
  
  // intersect account and paymaster ranges.
  function _intersectTimeRange(
      uint256 validationData,
      uint256 paymasterValidationData
  ) pure returns (ValidationData memory) {
      ValidationData memory accountValidationData = _parseValidationData(
          validationData
      );
      ValidationData memory pmValidationData = _parseValidationData(
          paymasterValidationData
      );
      address aggregator = accountValidationData.aggregator;
      if (aggregator == address(0)) {
          aggregator = pmValidationData.aggregator;
      }
      uint48 validAfter = accountValidationData.validAfter;
      uint48 validUntil = accountValidationData.validUntil;
      uint48 pmValidAfter = pmValidationData.validAfter;
      uint48 pmValidUntil = pmValidationData.validUntil;
  
      if (validAfter < pmValidAfter) validAfter = pmValidAfter;
      if (validUntil > pmValidUntil) validUntil = pmValidUntil;
      return ValidationData(aggregator, validAfter, validUntil);
  }
  
  /**
   * helper to pack the return value for validateUserOp
   * @param data - the ValidationData to pack
   */
  function _packValidationData(
      ValidationData memory data
  ) pure returns (uint256) {
      return
          uint160(data.aggregator) |
          (uint256(data.validUntil) << 160) |
          (uint256(data.validAfter) << (160 + 48));
  }
  
  /**
   * helper to pack the return value for validateUserOp, when not using an aggregator
   * @param sigFailed - true for signature failure, false for success
   * @param validUntil last timestamp this UserOperation is valid (or zero for infinite)
   * @param validAfter first timestamp this UserOperation is valid
   */
  function _packValidationData(
      bool sigFailed,
      uint48 validUntil,
      uint48 validAfter
  ) pure returns (uint256) {
      return
          (sigFailed ? 1 : 0) |
          (uint256(validUntil) << 160) |
          (uint256(validAfter) << (160 + 48));
  }
  
  /**
   * keccak function over calldata.
   * @dev copy calldata into memory, do keccak and drop allocated memory. Strangely, this is more efficient than letting solidity do it.
   */
  function calldataKeccak(bytes calldata data) pure returns (bytes32 ret) {
      assembly {
          let mem := mload(0x40)
          let len := data.length
          calldatacopy(mem, data.offset, len)
          ret := keccak256(mem, len)
      }
  }
  
  // File @account-abstraction/contracts/interfaces/UserOperation.sol@v0.6.0
  
  /**
   * User Operation struct
   * @param sender the sender account of this request.
   * @param nonce unique value the sender uses to verify it is not a replay.
   * @param initCode if set, the account contract will be created by this constructor/
   * @param callData the method call to execute on this account.
   * @param callGasLimit the gas limit passed to the callData method call.
   * @param verificationGasLimit gas used for validateUserOp and validatePaymasterUserOp.
   * @param preVerificationGas gas not calculated by the handleOps method, but added to the gas paid. Covers batch overhead.
   * @param maxFeePerGas same as EIP-1559 gas parameter.
   * @param maxPriorityFeePerGas same as EIP-1559 gas parameter.
   * @param paymasterAndData if set, this field holds the paymaster address and paymaster-specific data. the paymaster will pay for the transaction instead of the sender.
   * @param signature sender-verified signature over the entire request, the EntryPoint address and the chain ID.
   */
  struct UserOperation {
      address sender;
      uint256 nonce;
      bytes initCode;
      bytes callData;
      uint256 callGasLimit;
      uint256 verificationGasLimit;
      uint256 preVerificationGas;
      uint256 maxFeePerGas;
      uint256 maxPriorityFeePerGas;
      bytes paymasterAndData;
      bytes signature;
  }
  
  /**
   * Utility functions helpful when working with UserOperation structs.
   */
  library UserOperationLib {
      function getSender(
          UserOperation calldata userOp
      ) internal pure returns (address) {
          address data;
          //read sender from userOp, which is first userOp member (saves 800 gas...)
          assembly {
              data := calldataload(userOp)
          }
          return address(uint160(data));
      }
  
      //relayer/block builder might submit the TX with higher priorityFee, but the user should not
      // pay above what he signed for.
      function gasPrice(
          UserOperation calldata userOp
      ) internal view returns (uint256) {
          unchecked {
              uint256 maxFeePerGas = userOp.maxFeePerGas;
              uint256 maxPriorityFeePerGas = userOp.maxPriorityFeePerGas;
              if (maxFeePerGas == maxPriorityFeePerGas) {
                  //legacy mode (for networks that don't support basefee opcode)
                  return maxFeePerGas;
              }
              return min(maxFeePerGas, maxPriorityFeePerGas + block.basefee);
          }
      }
  
      function pack(
          UserOperation calldata userOp
      ) internal pure returns (bytes memory ret) {
          address sender = getSender(userOp);
          uint256 nonce = userOp.nonce;
          bytes32 hashInitCode = calldataKeccak(userOp.initCode);
          bytes32 hashCallData = calldataKeccak(userOp.callData);
          uint256 callGasLimit = userOp.callGasLimit;
          uint256 verificationGasLimit = userOp.verificationGasLimit;
          uint256 preVerificationGas = userOp.preVerificationGas;
          uint256 maxFeePerGas = userOp.maxFeePerGas;
          uint256 maxPriorityFeePerGas = userOp.maxPriorityFeePerGas;
          bytes32 hashPaymasterAndData = calldataKeccak(userOp.paymasterAndData);
  
          return
              abi.encode(
                  sender,
                  nonce,
                  hashInitCode,
                  hashCallData,
                  callGasLimit,
                  verificationGasLimit,
                  preVerificationGas,
                  maxFeePerGas,
                  maxPriorityFeePerGas,
                  hashPaymasterAndData
              );
      }
  
      function hash(
          UserOperation calldata userOp
      ) internal pure returns (bytes32) {
          return keccak256(pack(userOp));
      }
  
      function min(uint256 a, uint256 b) internal pure returns (uint256) {
          return a < b ? a : b;
      }
  }
  
  // File @openzeppelin/contracts/utils/math/Math.sol@v4.9.3
  
  /**
   * @dev Standard math utilities missing in the Solidity language.
   */
  library Math {
      enum Rounding {
          Down, // Toward negative infinity
          Up, // Toward infinity
          Zero // Toward zero
      }
  
      /**
       * @dev Returns the largest of two numbers.
       */
      function max(uint256 a, uint256 b) internal pure returns (uint256) {
          return a > b ? a : b;
      }
  
      /**
       * @dev Returns the smallest of two numbers.
       */
      function min(uint256 a, uint256 b) internal pure returns (uint256) {
          return a < b ? a : b;
      }
  
      /**
       * @dev Returns the average of two numbers. The result is rounded towards
       * zero.
       */
      function average(uint256 a, uint256 b) internal pure returns (uint256) {
          // (a + b) / 2 can overflow.
          return (a & b) + (a ^ b) / 2;
      }
  
      /**
       * @dev Returns the ceiling of the division of two numbers.
       *
       * This differs from standard division with `/` in that it rounds up instead
       * of rounding down.
       */
      function ceilDiv(uint256 a, uint256 b) internal pure returns (uint256) {
          // (a + b - 1) / b can overflow on addition, so we distribute.
          return a == 0 ? 0 : (a - 1) / b + 1;
      }
  
      /**
       * @notice Calculates floor(x * y / denominator) with full precision. Throws if result overflows a uint256 or denominator == 0
       * @dev Original credit to Remco Bloemen under MIT license (https://xn--2-umb.com/21/muldiv)
       * with further edits by Uniswap Labs also under MIT license.
       */
      function mulDiv(
          uint256 x,
          uint256 y,
          uint256 denominator
      ) internal pure returns (uint256 result) {
          unchecked {
              // 512-bit multiply [prod1 prod0] = x * y. Compute the product mod 2^256 and mod 2^256 - 1, then use
              // use the Chinese Remainder Theorem to reconstruct the 512 bit result. The result is stored in two 256
              // variables such that product = prod1 * 2^256 + prod0.
              uint256 prod0; // Least significant 256 bits of the product
              uint256 prod1; // Most significant 256 bits of the product
              assembly {
                  let mm := mulmod(x, y, not(0))
                  prod0 := mul(x, y)
                  prod1 := sub(sub(mm, prod0), lt(mm, prod0))
              }
  
              // Handle non-overflow cases, 256 by 256 division.
              if (prod1 == 0) {
                  // Solidity will revert if denominator == 0, unlike the div opcode on its own.
                  // The surrounding unchecked block does not change this fact.
                  // See https://docs.soliditylang.org/en/latest/control-structures.html#checked-or-unchecked-arithmetic.
                  return prod0 / denominator;
              }
  
              // Make sure the result is less than 2^256. Also prevents denominator == 0.
              require(denominator > prod1, "Math: mulDiv overflow");
  
              ///////////////////////////////////////////////
              // 512 by 256 division.
              ///////////////////////////////////////////////
  
              // Make division exact by subtracting the remainder from [prod1 prod0].
              uint256 remainder;
              assembly {
                  // Compute remainder using mulmod.
                  remainder := mulmod(x, y, denominator)
  
                  // Subtract 256 bit number from 512 bit number.
                  prod1 := sub(prod1, gt(remainder, prod0))
                  prod0 := sub(prod0, remainder)
              }
  
              // Factor powers of two out of denominator and compute largest power of two divisor of denominator. Always >= 1.
              // See https://cs.stackexchange.com/q/138556/92363.
  
              // Does not overflow because the denominator cannot be zero at this stage in the function.
              uint256 twos = denominator & (~denominator + 1);
              assembly {
                  // Divide denominator by twos.
                  denominator := div(denominator, twos)
  
                  // Divide [prod1 prod0] by twos.
                  prod0 := div(prod0, twos)
  
                  // Flip twos such that it is 2^256 / twos. If twos is zero, then it becomes one.
                  twos := add(div(sub(0, twos), twos), 1)
              }
  
              // Shift in bits from prod1 into prod0.
              prod0 |= prod1 * twos;
  
              // Invert denominator mod 2^256. Now that denominator is an odd number, it has an inverse modulo 2^256 such
              // that denominator * inv = 1 mod 2^256. Compute the inverse by starting with a seed that is correct for
              // four bits. That is, denominator * inv = 1 mod 2^4.
              uint256 inverse = (3 * denominator) ^ 2;
  
              // Use the Newton-Raphson iteration to improve the precision. Thanks to Hensel's lifting lemma, this also works
              // in modular arithmetic, doubling the correct bits in each step.
              inverse *= 2 - denominator * inverse; // inverse mod 2^8
              inverse *= 2 - denominator * inverse; // inverse mod 2^16
              inverse *= 2 - denominator * inverse; // inverse mod 2^32
              inverse *= 2 - denominator * inverse; // inverse mod 2^64
              inverse *= 2 - denominator * inverse; // inverse mod 2^128
              inverse *= 2 - denominator * inverse; // inverse mod 2^256
  
              // Because the division is now exact we can divide by multiplying with the modular inverse of denominator.
              // This will give us the correct result modulo 2^256. Since the preconditions guarantee that the outcome is
              // less than 2^256, this is the final result. We don't need to compute the high bits of the result and prod1
              // is no longer required.
              result = prod0 * inverse;
              return result;
          }
      }
  
      /**
       * @notice Calculates x * y / denominator with full precision, following the selected rounding direction.
       */
      function mulDiv(
          uint256 x,
          uint256 y,
          uint256 denominator,
          Rounding rounding
      ) internal pure returns (uint256) {
          uint256 result = mulDiv(x, y, denominator);
          if (rounding == Rounding.Up && mulmod(x, y, denominator) > 0) {
              result += 1;
          }
          return result;
      }
  
      /**
       * @dev Returns the square root of a number. If the number is not a perfect square, the value is rounded down.
       *
       * Inspired by Henry S. Warren, Jr.'s "Hacker's Delight" (Chapter 11).
       */
      function sqrt(uint256 a) internal pure returns (uint256) {
          if (a == 0) {
              return 0;
          }
  
          // For our first guess, we get the biggest power of 2 which is smaller than the square root of the target.
          //
          // We know that the "msb" (most significant bit) of our target number `a` is a power of 2 such that we have
          // `msb(a) <= a < 2*msb(a)`. This value can be written `msb(a)=2**k` with `k=log2(a)`.
          //
          // This can be rewritten `2**log2(a) <= a < 2**(log2(a) + 1)`
          // → `sqrt(2**k) <= sqrt(a) < sqrt(2**(k+1))`
          // → `2**(k/2) <= sqrt(a) < 2**((k+1)/2) <= 2**(k/2 + 1)`
          //
          // Consequently, `2**(log2(a) / 2)` is a good first approximation of `sqrt(a)` with at least 1 correct bit.
          uint256 result = 1 << (log2(a) >> 1);
  
          // At this point `result` is an estimation with one bit of precision. We know the true value is a uint128,
          // since it is the square root of a uint256. Newton's method converges quadratically (precision doubles at
          // every iteration). We thus need at most 7 iteration to turn our partial result with one bit of precision
          // into the expected uint128 result.
          unchecked {
              result = (result + a / result) >> 1;
              result = (result + a / result) >> 1;
              result = (result + a / result) >> 1;
              result = (result + a / result) >> 1;
              result = (result + a / result) >> 1;
              result = (result + a / result) >> 1;
              result = (result + a / result) >> 1;
              return min(result, a / result);
          }
      }
  
      /**
       * @notice Calculates sqrt(a), following the selected rounding direction.
       */
      function sqrt(
          uint256 a,
          Rounding rounding
      ) internal pure returns (uint256) {
          unchecked {
              uint256 result = sqrt(a);
              return
                  result +
                  (rounding == Rounding.Up && result * result < a ? 1 : 0);
          }
      }
  
      /**
       * @dev Return the log in base 2, rounded down, of a positive value.
       * Returns 0 if given 0.
       */
      function log2(uint256 value) internal pure returns (uint256) {
          uint256 result = 0;
          unchecked {
              if (value >> 128 > 0) {
                  value >>= 128;
                  result += 128;
              }
              if (value >> 64 > 0) {
                  value >>= 64;
                  result += 64;
              }
              if (value >> 32 > 0) {
                  value >>= 32;
                  result += 32;
              }
              if (value >> 16 > 0) {
                  value >>= 16;
                  result += 16;
              }
              if (value >> 8 > 0) {
                  value >>= 8;
                  result += 8;
              }
              if (value >> 4 > 0) {
                  value >>= 4;
                  result += 4;
              }
              if (value >> 2 > 0) {
                  value >>= 2;
                  result += 2;
              }
              if (value >> 1 > 0) {
                  result += 1;
              }
          }
          return result;
      }
  
      /**
       * @dev Return the log in base 2, following the selected rounding direction, of a positive value.
       * Returns 0 if given 0.
       */
      function log2(
          uint256 value,
          Rounding rounding
      ) internal pure returns (uint256) {
          unchecked {
              uint256 result = log2(value);
              return
                  result +
                  (rounding == Rounding.Up && 1 << result < value ? 1 : 0);
          }
      }
  
      /**
       * @dev Return the log in base 10, rounded down, of a positive value.
       * Returns 0 if given 0.
       */
      function log10(uint256 value) internal pure returns (uint256) {
          uint256 result = 0;
          unchecked {
              if (value >= 10 ** 64) {
                  value /= 10 ** 64;
                  result += 64;
              }
              if (value >= 10 ** 32) {
                  value /= 10 ** 32;
                  result += 32;
              }
              if (value >= 10 ** 16) {
                  value /= 10 ** 16;
                  result += 16;
              }
              if (value >= 10 ** 8) {
                  value /= 10 ** 8;
                  result += 8;
              }
              if (value >= 10 ** 4) {
                  value /= 10 ** 4;
                  result += 4;
              }
              if (value >= 10 ** 2) {
                  value /= 10 ** 2;
                  result += 2;
              }
              if (value >= 10 ** 1) {
                  result += 1;
              }
          }
          return result;
      }
  
      /**
       * @dev Return the log in base 10, following the selected rounding direction, of a positive value.
       * Returns 0 if given 0.
       */
      function log10(
          uint256 value,
          Rounding rounding
      ) internal pure returns (uint256) {
          unchecked {
              uint256 result = log10(value);
              return
                  result +
                  (rounding == Rounding.Up && 10 ** result < value ? 1 : 0);
          }
      }
  
      /**
       * @dev Return the log in base 256, rounded down, of a positive value.
       * Returns 0 if given 0.
       *
       * Adding one to the result gives the number of pairs of hex symbols needed to represent `value` as a hex string.
       */
      function log256(uint256 value) internal pure returns (uint256) {
          uint256 result = 0;
          unchecked {
              if (value >> 128 > 0) {
                  value >>= 128;
                  result += 16;
              }
              if (value >> 64 > 0) {
                  value >>= 64;
                  result += 8;
              }
              if (value >> 32 > 0) {
                  value >>= 32;
                  result += 4;
              }
              if (value >> 16 > 0) {
                  value >>= 16;
                  result += 2;
              }
              if (value >> 8 > 0) {
                  result += 1;
              }
          }
          return result;
      }
  
      /**
       * @dev Return the log in base 256, following the selected rounding direction, of a positive value.
       * Returns 0 if given 0.
       */
      function log256(
          uint256 value,
          Rounding rounding
      ) internal pure returns (uint256) {
          unchecked {
              uint256 result = log256(value);
              return
                  result +
                  (rounding == Rounding.Up && 1 << (result << 3) < value ? 1 : 0);
          }
      }
  }
  
  // File @openzeppelin/contracts/utils/math/SignedMath.sol@v4.9.3
  
  /**
   * @dev Standard signed math utilities missing in the Solidity language.
   */
  library SignedMath {
      /**
       * @dev Returns the largest of two signed numbers.
       */
      function max(int256 a, int256 b) internal pure returns (int256) {
          return a > b ? a : b;
      }
  
      /**
       * @dev Returns the smallest of two signed numbers.
       */
      function min(int256 a, int256 b) internal pure returns (int256) {
          return a < b ? a : b;
      }
  
      /**
       * @dev Returns the average of two signed numbers without overflow.
       * The result is rounded towards zero.
       */
      function average(int256 a, int256 b) internal pure returns (int256) {
          // Formula from the book "Hacker's Delight"
          int256 x = (a & b) + ((a ^ b) >> 1);
          return x + (int256(uint256(x) >> 255) & (a ^ b));
      }
  
      /**
       * @dev Returns the absolute unsigned value of a signed value.
       */
      function abs(int256 n) internal pure returns (uint256) {
          unchecked {
              // must be unchecked in order to support `n = type(int256).min`
              return uint256(n >= 0 ? n : -n);
          }
      }
  }
  
  // File @openzeppelin/contracts/utils/Strings.sol@v4.9.3
  
  /**
   * @dev String operations.
   */
  library Strings {
      bytes16 private constant _SYMBOLS = "0123456789abcdef";
      uint8 private constant _ADDRESS_LENGTH = 20;
  
      /**
       * @dev Converts a `uint256` to its ASCII `string` decimal representation.
       */
      function toString(uint256 value) internal pure returns (string memory) {
          unchecked {
              uint256 length = Math.log10(value) + 1;
              string memory buffer = new string(length);
              uint256 ptr;
              /// @solidity memory-safe-assembly
              assembly {
                  ptr := add(buffer, add(32, length))
              }
              while (true) {
                  ptr--;
                  /// @solidity memory-safe-assembly
                  assembly {
                      mstore8(ptr, byte(mod(value, 10), _SYMBOLS))
                  }
                  value /= 10;
                  if (value == 0) break;
              }
              return buffer;
          }
      }
  
      /**
       * @dev Converts a `int256` to its ASCII `string` decimal representation.
       */
      function toString(int256 value) internal pure returns (string memory) {
          return
              string(
                  abi.encodePacked(
                      value < 0 ? "-" : "",
                      toString(SignedMath.abs(value))
                  )
              );
      }
  
      /**
       * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
       */
      function toHexString(uint256 value) internal pure returns (string memory) {
          unchecked {
              return toHexString(value, Math.log256(value) + 1);
          }
      }
  
      /**
       * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
       */
      function toHexString(
          uint256 value,
          uint256 length
      ) internal pure returns (string memory) {
          bytes memory buffer = new bytes(2 * length + 2);
          buffer[0] = "0";
          buffer[1] = "x";
          for (uint256 i = 2 * length + 1; i > 1; --i) {
              buffer[i] = _SYMBOLS[value & 0xf];
              value >>= 4;
          }
          require(value == 0, "Strings: hex length insufficient");
          return string(buffer);
      }
  
      /**
       * @dev Converts an `address` with fixed length of 20 bytes to its not checksummed ASCII `string` hexadecimal representation.
       */
      function toHexString(address addr) internal pure returns (string memory) {
          return toHexString(uint256(uint160(addr)), _ADDRESS_LENGTH);
      }
  
      /**
       * @dev Returns true if the two strings are equal.
       */
      function equal(
          string memory a,
          string memory b
      ) internal pure returns (bool) {
          return keccak256(bytes(a)) == keccak256(bytes(b));
      }
  }
  
  // File @openzeppelin/contracts/utils/cryptography/ECDSA.sol@v4.9.3
  
  /**
   * @dev Elliptic Curve Digital Signature Algorithm (ECDSA) operations.
   *
   * These functions can be used to verify that a message was signed by the holder
   * of the private keys of a given address.
   */
  library ECDSA {
      enum RecoverError {
          NoError,
          InvalidSignature,
          InvalidSignatureLength,
          InvalidSignatureS,
          InvalidSignatureV // Deprecated in v4.8
      }
  
      function _throwError(RecoverError error) private pure {
          if (error == RecoverError.NoError) {
              return; // no error: do nothing
          } else if (error == RecoverError.InvalidSignature) {
              revert("ECDSA: invalid signature");
          } else if (error == RecoverError.InvalidSignatureLength) {
              revert("ECDSA: invalid signature length");
          } else if (error == RecoverError.InvalidSignatureS) {
              revert("ECDSA: invalid signature 's' value");
          }
      }
  
      /**
       * @dev Returns the address that signed a hashed message (`hash`) with
       * `signature` or error string. This address can then be used for verification purposes.
       *
       * The `ecrecover` EVM opcode allows for malleable (non-unique) signatures:
       * this function rejects them by requiring the `s` value to be in the lower
       * half order, and the `v` value to be either 27 or 28.
       *
       * IMPORTANT: `hash` _must_ be the result of a hash operation for the
       * verification to be secure: it is possible to craft signatures that
       * recover to arbitrary addresses for non-hashed data. A safe way to ensure
       * this is by receiving a hash of the original message (which may otherwise
       * be too long), and then calling {toEthSignedMessageHash} on it.
       *
       * Documentation for signature generation:
       * - with https://web3js.readthedocs.io/en/v1.3.4/web3-eth-accounts.html#sign[Web3.js]
       * - with https://docs.ethers.io/v5/api/signer/#Signer-signMessage[ethers]
       *
       * _Available since v4.3._
       */
      function tryRecover(
          bytes32 hash,
          bytes memory signature
      ) internal pure returns (address, RecoverError) {
          if (signature.length == 65) {
              bytes32 r;
              bytes32 s;
              uint8 v;
              // ecrecover takes the signature parameters, and the only way to get them
              // currently is to use assembly.
              /// @solidity memory-safe-assembly
              assembly {
                  r := mload(add(signature, 0x20))
                  s := mload(add(signature, 0x40))
                  v := byte(0, mload(add(signature, 0x60)))
              }
              return tryRecover(hash, v, r, s);
          } else {
              return (address(0), RecoverError.InvalidSignatureLength);
          }
      }
  
      /**
       * @dev Returns the address that signed a hashed message (`hash`) with
       * `signature`. This address can then be used for verification purposes.
       *
       * The `ecrecover` EVM opcode allows for malleable (non-unique) signatures:
       * this function rejects them by requiring the `s` value to be in the lower
       * half order, and the `v` value to be either 27 or 28.
       *
       * IMPORTANT: `hash` _must_ be the result of a hash operation for the
       * verification to be secure: it is possible to craft signatures that
       * recover to arbitrary addresses for non-hashed data. A safe way to ensure
       * this is by receiving a hash of the original message (which may otherwise
       * be too long), and then calling {toEthSignedMessageHash} on it.
       */
      function recover(
          bytes32 hash,
          bytes memory signature
      ) internal pure returns (address) {
          (address recovered, RecoverError error) = tryRecover(hash, signature);
          _throwError(error);
          return recovered;
      }
  
      /**
       * @dev Overload of {ECDSA-tryRecover} that receives the `r` and `vs` short-signature fields separately.
       *
       * See https://eips.ethereum.org/EIPS/eip-2098[EIP-2098 short signatures]
       *
       * _Available since v4.3._
       */
      function tryRecover(
          bytes32 hash,
          bytes32 r,
          bytes32 vs
      ) internal pure returns (address, RecoverError) {
          bytes32 s = vs &
              bytes32(
                  0x7fffffffffffffffffffffffffffffffffffffffffffffffffffffffffffffff
              );
          uint8 v = uint8((uint256(vs) >> 255) + 27);
          return tryRecover(hash, v, r, s);
      }
  
      /**
       * @dev Overload of {ECDSA-recover} that receives the `r and `vs` short-signature fields separately.
       *
       * _Available since v4.2._
       */
      function recover(
          bytes32 hash,
          bytes32 r,
          bytes32 vs
      ) internal pure returns (address) {
          (address recovered, RecoverError error) = tryRecover(hash, r, vs);
          _throwError(error);
          return recovered;
      }
  
      /**
       * @dev Overload of {ECDSA-tryRecover} that receives the `v`,
       * `r` and `s` signature fields separately.
       *
       * _Available since v4.3._
       */
      function tryRecover(
          bytes32 hash,
          uint8 v,
          bytes32 r,
          bytes32 s
      ) internal pure returns (address, RecoverError) {
          // EIP-2 still allows signature malleability for ecrecover(). Remove this possibility and make the signature
          // unique. Appendix F in the Ethereum Yellow paper (https://ethereum.github.io/yellowpaper/paper.pdf), defines
          // the valid range for s in (301): 0 < s < secp256k1n ÷ 2 + 1, and for v in (302): v ∈ {27, 28}. Most
          // signatures from current libraries generate a unique signature with an s-value in the lower half order.
          //
          // If your library generates malleable signatures, such as s-values in the upper range, calculate a new s-value
          // with 0xFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFEBAAEDCE6AF48A03BBFD25E8CD0364141 - s1 and flip v from 27 to 28 or
          // vice versa. If your library also generates signatures with 0/1 for v instead 27/28, add 27 to v to accept
          // these malleable signatures as well.
          if (
              uint256(s) >
              0x7FFFFFFFFFFFFFFFFFFFFFFFFFFFFFFF5D576E7357A4501DDFE92F46681B20A0
          ) {
              return (address(0), RecoverError.InvalidSignatureS);
          }
  
          // If the signature is valid (and not malleable), return the signer address
          address signer = ecrecover(hash, v, r, s);
          if (signer == address(0)) {
              return (address(0), RecoverError.InvalidSignature);
          }
  
          return (signer, RecoverError.NoError);
      }
  
      /**
       * @dev Overload of {ECDSA-recover} that receives the `v`,
       * `r` and `s` signature fields separately.
       */
      function recover(
          bytes32 hash,
          uint8 v,
          bytes32 r,
          bytes32 s
      ) internal pure returns (address) {
          (address recovered, RecoverError error) = tryRecover(hash, v, r, s);
          _throwError(error);
          return recovered;
      }
  
      /**
       * @dev Returns an Ethereum Signed Message, created from a `hash`. This
       * produces hash corresponding to the one signed with the
       * https://eth.wiki/json-rpc/API#eth_sign[`eth_sign`]
       * JSON-RPC method as part of EIP-191.
       *
       * See {recover}.
       */
      function toEthSignedMessageHash(
          bytes32 hash
      ) internal pure returns (bytes32 message) {
          // 32 is the length in bytes of hash,
          // enforced by the type signature above
          /// @solidity memory-safe-assembly
          assembly {
              mstore(0x00, "\x19Ethereum Signed Message:\n32")
              mstore(0x1c, hash)
              message := keccak256(0x00, 0x3c)
          }
      }
  
      /**
       * @dev Returns an Ethereum Signed Message, created from `s`. This
       * produces hash corresponding to the one signed with the
       * https://eth.wiki/json-rpc/API#eth_sign[`eth_sign`]
       * JSON-RPC method as part of EIP-191.
       *
       * See {recover}.
       */
      function toEthSignedMessageHash(
          bytes memory s
      ) internal pure returns (bytes32) {
          return
              keccak256(
                  abi.encodePacked(
                      "\x19Ethereum Signed Message:\n",
                      Strings.toString(s.length),
                      s
                  )
              );
      }
  
      /**
       * @dev Returns an Ethereum Signed Typed Data, created from a
       * `domainSeparator` and a `structHash`. This produces hash corresponding
       * to the one signed with the
       * https://eips.ethereum.org/EIPS/eip-712[`eth_signTypedData`]
       * JSON-RPC method as part of EIP-712.
       *
       * See {recover}.
       */
      function toTypedDataHash(
          bytes32 domainSeparator,
          bytes32 structHash
      ) internal pure returns (bytes32 data) {
          /// @solidity memory-safe-assembly
          assembly {
              let ptr := mload(0x40)
              mstore(ptr, "\x19\x01")
              mstore(add(ptr, 0x02), domainSeparator)
              mstore(add(ptr, 0x22), structHash)
              data := keccak256(ptr, 0x42)
          }
      }
  
      /**
       * @dev Returns an Ethereum Signed Data with intended validator, created from a
       * `validator` and `data` according to the version 0 of EIP-191.
       *
       * See {recover}.
       */
      function toDataWithIntendedValidatorHash(
          address validator,
          bytes memory data
      ) internal pure returns (bytes32) {
          return keccak256(abi.encodePacked("\x19\x00", validator, data));
      }
  }
  
  // File biconomy/contracts/smart-account/interfaces/IAuthorizationModule.sol
  
  // interface for modules to verify singatures signed over userOpHash
  interface IAuthorizationModule {
      function validateUserOp(
          UserOperation calldata userOp,
          bytes32 userOpHash
      ) external returns (uint256 validationData);
  }
  
  // File biconomy/contracts/smart-account/interfaces/ISignatureValidator.sol
  
  contract ISignatureValidatorConstants {
      // bytes4(keccak256("isValidSignature(bytes32,bytes)")
      bytes4 internal constant EIP1271_MAGIC_VALUE = 0x1626ba7e;
  }
  
  abstract contract ISignatureValidator is ISignatureValidatorConstants {
      /**
       * @dev Should return whether the signature provided is valid for the provided data
       * @param _dataHash Arbitrary length data signed on behalf of address(this)
       * @param _signature Signature byte array associated with _data
       *
       * MUST return the bytes4 magic value 0x1626ba7e when function passes.
       * MUST NOT modify state (using STATICCALL for solc < 0.5, view modifier for solc > 0.5)
       * MUST allow external calls
       */
      function isValidSignature(
          bytes32 _dataHash,
          bytes memory _signature
      ) public view virtual returns (bytes4);
  }
  
  // File biconomy/contracts/smart-account/modules/BaseAuthorizationModule.sol
  
  contract AuthorizationModulesConstants {
      uint256 internal constant VALIDATION_SUCCESS = 0;
      uint256 internal constant SIG_VALIDATION_FAILED = 1;
  }
  
  abstract contract BaseAuthorizationModule is
      IAuthorizationModule,
      ISignatureValidator,
      AuthorizationModulesConstants
  {}
  
  // File @openzeppelin/contracts/utils/cryptography/MerkleProof.sol@v4.9.3
  
  /**
   * @dev These functions deal with verification of Merkle Tree proofs.
   *
   * The tree and the proofs can be generated using our
   * https://github.com/OpenZeppelin/merkle-tree[JavaScript library].
   * You will find a quickstart guide in the readme.
   *
   * WARNING: You should avoid using leaf values that are 64 bytes long prior to
   * hashing, or use a hash function other than keccak256 for hashing leaves.
   * This is because the concatenation of a sorted pair of internal nodes in
   * the merkle tree could be reinterpreted as a leaf value.
   * OpenZeppelin's JavaScript library generates merkle trees that are safe
   * against this attack out of the box.
   */
  library MerkleProof {
      /**
       * @dev Returns true if a `leaf` can be proved to be a part of a Merkle tree
       * defined by `root`. For this, a `proof` must be provided, containing
       * sibling hashes on the branch from the leaf to the root of the tree. Each
       * pair of leaves and each pair of pre-images are assumed to be sorted.
       */
      function verify(
          bytes32[] memory proof,
          bytes32 root,
          bytes32 leaf
      ) internal pure returns (bool) {
          return processProof(proof, leaf) == root;
      }
  
      /**
       * @dev Calldata version of {verify}
       *
       * _Available since v4.7._
       */
      function verifyCalldata(
          bytes32[] calldata proof,
          bytes32 root,
          bytes32 leaf
      ) internal pure returns (bool) {
          return processProofCalldata(proof, leaf) == root;
      }
  
      /**
       * @dev Returns the rebuilt hash obtained by traversing a Merkle tree up
       * from `leaf` using `proof`. A `proof` is valid if and only if the rebuilt
       * hash matches the root of the tree. When processing the proof, the pairs
       * of leafs & pre-images are assumed to be sorted.
       *
       * _Available since v4.4._
       */
      function processProof(
          bytes32[] memory proof,
          bytes32 leaf
      ) internal pure returns (bytes32) {
          bytes32 computedHash = leaf;
          for (uint256 i = 0; i < proof.length; i++) {
              computedHash = _hashPair(computedHash, proof[i]);
          }
          return computedHash;
      }
  
      /**
       * @dev Calldata version of {processProof}
       *
       * _Available since v4.7._
       */
      function processProofCalldata(
          bytes32[] calldata proof,
          bytes32 leaf
      ) internal pure returns (bytes32) {
          bytes32 computedHash = leaf;
          for (uint256 i = 0; i < proof.length; i++) {
              computedHash = _hashPair(computedHash, proof[i]);
          }
          return computedHash;
      }
  
      /**
       * @dev Returns true if the `leaves` can be simultaneously proven to be a part of a merkle tree defined by
       * `root`, according to `proof` and `proofFlags` as described in {processMultiProof}.
       *
       * CAUTION: Not all merkle trees admit multiproofs. See {processMultiProof} for details.
       *
       * _Available since v4.7._
       */
      function multiProofVerify(
          bytes32[] memory proof,
          bool[] memory proofFlags,
          bytes32 root,
          bytes32[] memory leaves
      ) internal pure returns (bool) {
          return processMultiProof(proof, proofFlags, leaves) == root;
      }
  
      /**
       * @dev Calldata version of {multiProofVerify}
       *
       * CAUTION: Not all merkle trees admit multiproofs. See {processMultiProof} for details.
       *
       * _Available since v4.7._
       */
      function multiProofVerifyCalldata(
          bytes32[] calldata proof,
          bool[] calldata proofFlags,
          bytes32 root,
          bytes32[] memory leaves
      ) internal pure returns (bool) {
          return processMultiProofCalldata(proof, proofFlags, leaves) == root;
      }
  
      /**
       * @dev Returns the root of a tree reconstructed from `leaves` and sibling nodes in `proof`. The reconstruction
       * proceeds by incrementally reconstructing all inner nodes by combining a leaf/inner node with either another
       * leaf/inner node or a proof sibling node, depending on whether each `proofFlags` item is true or false
       * respectively.
       *
       * CAUTION: Not all merkle trees admit multiproofs. To use multiproofs, it is sufficient to ensure that: 1) the tree
       * is complete (but not necessarily perfect), 2) the leaves to be proven are in the opposite order they are in the
       * tree (i.e., as seen from right to left starting at the deepest layer and continuing at the next layer).
       *
       * _Available since v4.7._
       */
      function processMultiProof(
          bytes32[] memory proof,
          bool[] memory proofFlags,
          bytes32[] memory leaves
      ) internal pure returns (bytes32 merkleRoot) {
          // This function rebuilds the root hash by traversing the tree up from the leaves. The root is rebuilt by
          // consuming and producing values on a queue. The queue starts with the `leaves` array, then goes onto the
          // `hashes` array. At the end of the process, the last hash in the `hashes` array should contain the root of
          // the merkle tree.
          uint256 leavesLen = leaves.length;
          uint256 proofLen = proof.length;
          uint256 totalHashes = proofFlags.length;
  
          // Check proof validity.
          require(
              leavesLen + proofLen - 1 == totalHashes,
              "MerkleProof: invalid multiproof"
          );
  
          // The xxxPos values are "pointers" to the next value to consume in each array. All accesses are done using
          // `xxx[xxxPos++]`, which return the current value and increment the pointer, thus mimicking a queue's "pop".
          bytes32[] memory hashes = new bytes32[](totalHashes);
          uint256 leafPos = 0;
          uint256 hashPos = 0;
          uint256 proofPos = 0;
          // At each step, we compute the next hash using two values:
          // - a value from the "main queue". If not all leaves have been consumed, we get the next leaf, otherwise we
          //   get the next hash.
          // - depending on the flag, either another value from the "main queue" (merging branches) or an element from the
          //   `proof` array.
          for (uint256 i = 0; i < totalHashes; i++) {
              bytes32 a = leafPos < leavesLen
                  ? leaves[leafPos++]
                  : hashes[hashPos++];
              bytes32 b = proofFlags[i]
                  ? (leafPos < leavesLen ? leaves[leafPos++] : hashes[hashPos++])
                  : proof[proofPos++];
              hashes[i] = _hashPair(a, b);
          }
  
          if (totalHashes > 0) {
              require(proofPos == proofLen, "MerkleProof: invalid multiproof");
              unchecked {
                  return hashes[totalHashes - 1];
              }
          } else if (leavesLen > 0) {
              return leaves[0];
          } else {
              return proof[0];
          }
      }
  
      /**
       * @dev Calldata version of {processMultiProof}.
       *
       * CAUTION: Not all merkle trees admit multiproofs. See {processMultiProof} for details.
       *
       * _Available since v4.7._
       */
      function processMultiProofCalldata(
          bytes32[] calldata proof,
          bool[] calldata proofFlags,
          bytes32[] memory leaves
      ) internal pure returns (bytes32 merkleRoot) {
          // This function rebuilds the root hash by traversing the tree up from the leaves. The root is rebuilt by
          // consuming and producing values on a queue. The queue starts with the `leaves` array, then goes onto the
          // `hashes` array. At the end of the process, the last hash in the `hashes` array should contain the root of
          // the merkle tree.
          uint256 leavesLen = leaves.length;
          uint256 proofLen = proof.length;
          uint256 totalHashes = proofFlags.length;
  
          // Check proof validity.
          require(
              leavesLen + proofLen - 1 == totalHashes,
              "MerkleProof: invalid multiproof"
          );
  
          // The xxxPos values are "pointers" to the next value to consume in each array. All accesses are done using
          // `xxx[xxxPos++]`, which return the current value and increment the pointer, thus mimicking a queue's "pop".
          bytes32[] memory hashes = new bytes32[](totalHashes);
          uint256 leafPos = 0;
          uint256 hashPos = 0;
          uint256 proofPos = 0;
          // At each step, we compute the next hash using two values:
          // - a value from the "main queue". If not all leaves have been consumed, we get the next leaf, otherwise we
          //   get the next hash.
          // - depending on the flag, either another value from the "main queue" (merging branches) or an element from the
          //   `proof` array.
          for (uint256 i = 0; i < totalHashes; i++) {
              bytes32 a = leafPos < leavesLen
                  ? leaves[leafPos++]
                  : hashes[hashPos++];
              bytes32 b = proofFlags[i]
                  ? (leafPos < leavesLen ? leaves[leafPos++] : hashes[hashPos++])
                  : proof[proofPos++];
              hashes[i] = _hashPair(a, b);
          }
  
          if (totalHashes > 0) {
              require(proofPos == proofLen, "MerkleProof: invalid multiproof");
              unchecked {
                  return hashes[totalHashes - 1];
              }
          } else if (leavesLen > 0) {
              return leaves[0];
          } else {
              return proof[0];
          }
      }
  
      function _hashPair(bytes32 a, bytes32 b) private pure returns (bytes32) {
          return a < b ? _efficientHash(a, b) : _efficientHash(b, a);
      }
  
      function _efficientHash(
          bytes32 a,
          bytes32 b
      ) private pure returns (bytes32 value) {
          /// @solidity memory-safe-assembly
          assembly {
              mstore(0x00, a)
              mstore(0x20, b)
              value := keccak256(0x00, 0x40)
          }
      }
  }
  
  // File contracts/KlasterEcdsaModule.sol
  
  /**
   * @title Klaster ECDSA ownership Authorization module for Biconomy Smart Accounts compatible with the Klaster execution network.
   * @dev Compatible with Biconomy Modular Interface v 0.1
   *         - It allows to validate a part of iTx signed by EOA private key.
   *         - EIP-1271 compatible (ensures Smart Account can validate signed messages).
   *         - One owner per Smart Account.
   *         - Does not support outdated eth_sign flow for cheaper validations
   *         (see https://support.metamask.io/hc/en-us/articles/14764161421467-What-is-eth-sign-and-why-is-it-a-risk-)
   * !!!!!!! Only EOA owners supported, no Smart Account Owners
   *         For Smart Contract Owners check SmartContractOwnership module instead
   *
   */
  contract KlasterEcdsaModule is BaseAuthorizationModule {
      using ECDSA for bytes32;
      using UserOperationLib for UserOperation;
  
      string public constant NAME = "Klaster ECDSA Module";
      string public constant VERSION = "0.1.0";
      mapping(address => address) internal _smartAccountOwners;
  
      event OwnershipTransferred(
          address indexed smartAccount,
          address indexed oldOwner,
          address indexed newOwner
      );
  
      error NoOwnerRegisteredForSmartAccount(address smartAccount);
      error AlreadyInitedForSmartAccount(address smartAccount);
      error WrongSignatureLength();
      error NotEOA(address account);
      error ZeroAddressNotAllowedAsOwner();
  
      /**
       * @dev Initializes the module for a Smart Account.
       * Should be used at a time of first enabling the module for a Smart Account.
       * @param eoaOwner The owner of the Smart Account. Should be EOA!
       */
      function initForSmartAccount(address eoaOwner) external returns (address) {
          if (_smartAccountOwners[msg.sender] != address(0)) {
              revert AlreadyInitedForSmartAccount(msg.sender);
          }
          if (eoaOwner == address(0)) revert ZeroAddressNotAllowedAsOwner();
          if (_isSmartContract(eoaOwner)) revert NotEOA(eoaOwner);
          _smartAccountOwners[msg.sender] = eoaOwner;
          return address(this);
      }
  
      /**
       * @dev Sets/changes an for a Smart Account.
       * Should be called by Smart Account itself.
       * @param owner The owner of the Smart Account.
       */
      function transferOwnership(address owner) external {
          if (_isSmartContract(owner)) revert NotEOA(owner);
          if (owner == address(0)) revert ZeroAddressNotAllowedAsOwner();
          _transferOwnership(msg.sender, owner);
      }
  
      /**
       * @dev Renounces ownership
       * should be called by Smart Account.
       */
      function renounceOwnership() external {
          _transferOwnership(msg.sender, address(0));
      }
  
      /**
       * @dev Returns the owner of the Smart Account. Reverts for Smart Accounts without owners.
       * @param smartAccount Smart Account address.
       * @return owner The owner of the Smart Account.
       */
      function getOwner(address smartAccount) external view returns (address) {
          address owner = _smartAccountOwners[smartAccount];
          if (owner == address(0)) {
              revert NoOwnerRegisteredForSmartAccount(smartAccount);
          }
          return owner;
      }
  
      /**
       * Calculates userOp hash. Almost works like a regular 4337 userOp hash with few fields added.
       *
       * @param userOp userOp to calculate the hash for
       * @param lowerBoundTimestamp lower bound timestamp set when constructing userOp
       * @param upperBoundTimestamp upper bound timestamp set when constructing userOp
       */
      function getUserOpHash(
          UserOperation calldata userOp,
          uint256 lowerBoundTimestamp,
          uint256 upperBoundTimestamp
      ) public view returns (bytes32 userOpHash) {
          userOpHash = keccak256(
              bytes.concat(
                  keccak256(
                      abi.encode(
                          userOp.hash(),
                          lowerBoundTimestamp,
                          upperBoundTimestamp,
                          block.chainid
                      )
                  )
              )
          );
      }
  
      /**
       * @dev validates userOperation
       * @param userOp User Operation to be validated.
       * @param userOpHash Not used. Here for interface compatibility. iTxHash (merkle root) is used and encoded in userOp.signature
       * @return sigValidationResult ValidationData - success if:
       *                                  1. the given userOp belongs to the merkle tree with given iTxHash
       *                                  2. the given iTxHash has been signed by the owner of this smart account
       */
      function validateUserOp(
          UserOperation calldata userOp,
          bytes32 userOpHash
      ) external view virtual returns (uint256) {
          (bytes memory sigBytes, ) = abi.decode(
              userOp.signature,
              (bytes, address)
          );
  
          (
              bytes32 iTxHash,
              bytes32[] memory proof,
              uint48 lowerBoundTimestamp,
              uint48 upperBoundTimestamp,
              bytes memory userEcdsaSignature
          ) = abi.decode(sigBytes, (bytes32, bytes32[], uint48, uint48, bytes));
  
          bytes32 calculatedUserOpHash = getUserOpHash(
              userOp,
              lowerBoundTimestamp,
              upperBoundTimestamp
          );
          if (!_validateUserOpHash(calculatedUserOpHash, iTxHash, proof)) {
              return SIG_VALIDATION_FAILED;
          }
  
          if (!_verifySignature(iTxHash, userEcdsaSignature, userOp.sender)) {
              return SIG_VALIDATION_FAILED;
          }
  
          return
              _packValidationData(
                  false,
                  upperBoundTimestamp,
                  lowerBoundTimestamp
              );
      }
  
      /**
       * @dev Validates a signature for a message.
       * To be called from a Smart Account.
       * @param dataHash Exact hash of the data that was signed.
       * @param moduleSignature Signature to be validated.
       * @return EIP1271_MAGIC_VALUE if signature is valid, 0xffffffff otherwise.
       */
      function isValidSignature(
          bytes32 dataHash,
          bytes memory moduleSignature
      ) public view virtual override returns (bytes4) {
          return
              isValidSignatureForAddress(dataHash, moduleSignature, msg.sender);
      }
  
      /**
       * @dev Validates a signature for a message signed by address.
       * @dev Also try dataHash.toEthSignedMessageHash()
       * @param dataHash hash of the data
       * @param moduleSignature Signature to be validated.
       * @param smartAccount expected signer Smart Account address.
       * @return EIP1271_MAGIC_VALUE if signature is valid, 0xffffffff otherwise.
       */
      function isValidSignatureForAddress(
          bytes32 dataHash,
          bytes memory moduleSignature,
          address smartAccount
      ) public view virtual returns (bytes4) {
          if (_verifySignature(dataHash, moduleSignature, smartAccount)) {
              return EIP1271_MAGIC_VALUE;
          }
          return bytes4(0xffffffff);
      }
  
      /**
       * @dev Transfers ownership for smartAccount and emits an event
       * @param newOwner Smart Account address.
       */
      function _transferOwnership(
          address smartAccount,
          address newOwner
      ) internal {
          address _oldOwner = _smartAccountOwners[smartAccount];
          _smartAccountOwners[smartAccount] = newOwner;
          emit OwnershipTransferred(smartAccount, _oldOwner, newOwner);
      }
  
      function _validateUserOpHash(
          bytes32 userOpHash,
          bytes32 iTxHash,
          bytes32[] memory proof
      ) private pure returns (bool) {
          return MerkleProof.verify(proof, iTxHash, userOpHash);
      }
  
      /**
       * @dev Validates a signature for a message.
       * @dev Check if signature was made over dataHash.toEthSignedMessageHash() or just dataHash
       * The former is for personal_sign, the latter for the typed_data sign
       * Only EOA owners supported, no Smart Account Owners
       * For Smart Contract Owners check SmartContractOwnership Module instead
       * @param dataHash Hash of the data to be validated.
       * @param signature Signature to be validated.
       * @param smartAccount expected signer Smart Account address.
       * @return true if signature is valid, false otherwise.
       */
      function _verifySignature(
          bytes32 dataHash,
          bytes memory signature,
          address smartAccount
      ) internal view returns (bool) {
          address expectedSigner = _smartAccountOwners[smartAccount];
          if (expectedSigner == address(0)) {
              revert NoOwnerRegisteredForSmartAccount(smartAccount);
          }
          if (signature.length < 65) revert WrongSignatureLength();
          address recovered = (dataHash.toEthSignedMessageHash()).recover(
              signature
          );
          if (expectedSigner == recovered) {
              return true;
          }
          recovered = dataHash.recover(signature);
          if (expectedSigner == recovered) {
              return true;
          }
          return false;
      }
  
      /**
       * @dev Checks if the address provided is a smart contract.
       * @param account Address to be checked.
       */
      function _isSmartContract(address account) internal view returns (bool) {
          uint256 size;
          assembly {
              size := extcodesize(account)
          }
          return size > 0;
      }
  }